
Sono attributi dell’oggetto come i campi. Possono associare azioni alla lettura e alla scrittura dell’attributo

Sintassi:

property propertyName: type specifiers

· propertyName è un qualsiasi valido identificatore

· type è il tipo della proprietà

· specifiers è una sequenza di specificatori: read, write, stored, default (nodefault)

Specificatori di accesso

Ogni proprietà ha uno specificatore read, uno specificatore write o entrambi. Questi sono gli specificatori di accesso e hanno la seguente forma:

read campoOmetodo

write campoOmetodo

· Se l’argomento dello specificatore d’accesso è un campo, deve essere dello stesso tipo della proprietà 

· Se è un metodo non può essere overloaded 

· In un specificatore read il metodo deve essere una funzione che ritorna un valore dello stesso tipo 

· In uno specificatore write il metodo deve essere una procedura che ha un singolo parametro costante del tipo della proprietà

P.e.

property Color: TColor read GetColor write SetColor;

{ the GetColor method must be declared as }

function GetColor: TColor;

{ and the SetColor method must be declared as one of these: }

procedure SetColor(Value: TColor);

procedure SetColor(const Value: TColor);

Considerazioni

· Il valore della proprietà viene letto con il campo o il metodo specificato in read, viene scritto usando il campo o il metodo specificato in write. 

· Se non è specificato write è una proprietà read-only, se non è specificato read è una proprietà write-only.

· Al contrario dei campi i metodi non possono essere passati per riferimento e non gli si può applicare l’operatore @. Il motivo è che il valore di una proprietà può non esistere in memoria, per cui non ha nessun indirizzo.

Array properties

Sintassi:

Property propertyName[indexes]: type specifiers

Novità:

· [indexes] è una lista di indici racchiusa tra quadre

P.e.:

property Pixels[X, Y: Integer]: TColor read GetPixel write SetPixel;

property Values[const Name: string]: string read GetValue write SetValue;

· Le proprietà array possono avere indici di qualsiasi tipo

· Gli specificatori di accesso devono per forza listare metodi

· I metodi di accesso hanno in più la lista degli indici nello stesso ordine (il metodo in write ha “value” come ultimo parametro)

P.e.

function GetPixel(X, Y: Integer): TColor;

function GetValue(const Name: string): string;

procedure SetPixel(X, Y: Integer; Value: TColor);

procedure SetValue(const Name, Value: string);

· La direttiva default specifica la proprietà di default della classe. P.e. vedi TStrings.strings

Index specifiers

Permettono di condividere gli stessi metodi di accesso tra più proprietà della stessa classe

Sintassi:

Property propertyName: type index integerConstant specifiers

Novità:

· Index integerConstant è un indice intero che definisce la proprietà all’interno del metodo

P.e.:

    property Left: Longint index 0 read GetCoordinate write SetCoordinate;

…
    property Bottom: Longint index 3 read GetCoordinate write SetCoordinate;

    property Coordinates[Index: Integer]: Longint read GetCoordinate write SetCoordinate;

    ...

· In presenza di uno specificatore index gli specificatori read e write devono listare metodi

· I metodi di accesso devono avere un parametro intero extra: per una funzione read deve essere l’ultimo argomento, per la write deve essere il penultimo (prima del valore)

    function GetCoordinate(Index: Integer): Longint;

    procedure SetCoordinate(Index: Integer; Value: Longint);

Storage specifiers

Le direttive di immagazzinamento sono: stored, default, nodefault, tutte opzionali.

· Stored deve essere seguito da un valore booleano e indica se la proprietà deve essere salvata sul “.dfm”. Per default è true
· Default + valore specifica il valore di default, quello per cui non c’è bisogno di salvare la proprietà sul “.dfm”
· Nodefault specifica che non c’è nessun valore di default, che se stored è true la proprietà va sempre salvata
Considerazioni:

· Reali, puntatori e stringhe hanno i seguenti valori di default impliciti: 0, nil, ‘’
· Default e nodefault possono essere usati su proprietà di tipo ordinale, set, puntatore. Nodefault può essere usato anche su proprietà di tipo string e real.
· Le proprietà di tipo array non possono avere storage specifiers. Abbiamo visto che per le proprietà array default significa un’altra cosa

Property overrides

Una dichiarazione di proprietà senza tipo è chiamata property override. 

Un override di proprietà permette di:

· Incrementare la visibilità della proprietà

· Modificare le direttive di immagazzinamento

· Rimpiazzare o aggiungere uno specificatore di accesso

Non permette di:

· Diminuire la visibilità

· Togliere uno specificatore di accesso

P.e:

type

  TAncestor = class
  protected

    property Size: Integer read FSize;

    property Text: string read GetText write SetText;

    property Color: TColor read FColor write SetColor stored False;

    ...

  end;

  TDerived = class(TAncestor)

  protected

    property Size write SetSize;

  published

    property Text;

    property Color stored True default clBlue;

    ...

  end;

Property redeclarations

La ridichiarazione di una proprietà che include il tipo nasconde la proprietà ereditata, non la modifica. Questo significa che viene creata una nuova proprietà con lo stesso nome di quella vecchia

Attenzione:

· Sia nel caso di un property overriding, sia nel caso di un property redeclaration il binding di una proprietà è sempre statico

P.e.:

type

  TAncestor = class
     ...

    property Value: Integer read Method1 write Method2;

  end;

  TDescendant = class(TAncestor)

     ...

    property Value: Integer read Method3 write Method4;

  end;

var MyObject: TAncestor;

 ...

MyObject := TDescendant.Create; 

{ Access to Value through Method1,Method 2 }

Tipi procedurali

Le variabili di tipo procedurale contengono indirizzi di procedure e funzioni. Per dichiarare una variabile di tipo procedurale basta far seguire a “:” il prototipo delle funzioni a cui punta senza nessun identificatore.

P.e.:

function Calc(X,Y: Integer): Integer;

var F: function(X,Y: Integer): Integer; // procedural type
//You can assign the Calc function to the 

//variable F:

F := Calc;

Altri esempi:

type

  TIntegerFunction = function: Integer;

  TProcedure = procedure;

  TStrProc = procedure(const S: string);

  TMathFunc = function(X: Double): Double;

Per invocare una procedura o una funzione che è il metodo di una classe bisogna far seguire al prototipo le parole chiave of object

P.e.:

type

  TMethod = procedure of object;

  TNotifyEvent = procedure(Sender: TObject) of object;

Un puntatore a metodo in realtà ha 2 valori: il puntatore al codice e all’oggetto su cui deve operare (p.e. OnClick := MainForm.ButtonClick). Le procedure/funzioni nidificate non possono essere passate

Tipo “riferimento a classi”

Si possono dichiarare variabili il cui contenuto specifica un tipo di classe. Per farlo si utilizza la sintassi:

class of type

Per esempio:

type 

  TClass = class of TObject;

var

  MyObj: TClass; 

…

  MyObj := TButton;

Una variabile riferimento a classe può contenere un riferimento a una qualsiasi classe figlia del tipo a cui si riferisce. P.e.:

type 

  TClassButton = class of TButton;

var

  MyObj: TClassButton; 

…

  MyObj := TEdit; // errore

Considerazioni

· La definizione di un tipo riferimento a classe non può essere costruita nella dichiarazione di una variabile o di una lista di parametri

· Possono essere utili se un costruttore o un metodo hanno bisogno dell’indicazione di un tipo di classe dal chiamante. P.e.:

type TCollectionItemClass = class of TCollectionItem;

  ...

constructor Create(ItemClass: TCollectionItemClass);

Costruttori e riferimenti a classe

Un costruttore può essere invocato con una variabile di tipo riferimento a classe. In tal caso, se è definito con binding dinamico, permette la costruzione di oggetti il cui tipo non è conosciuto a tempo di compilazione. 

P.e.:

type TControlClass = class of TControl;

function CreateControl(ControlClass: TControlClass; const ControlName: string; X, Y, W, H: Integer): TControl;

begin

  Result := ControlClass.Create(MainForm);

  with Result do
  begin

    Parent := MainForm;

    Name := ControlName;

    SetBounds(X, Y, W, H);

    Visible := True;

  end;

end;

Osservazioni

· Si noti che grazie all’uso della variabile riferimento a classe è possibile invocare costruttori differenti: in parole povere invocare il costruttore in maniera dinamica

· Il trucco funziona solo se i costruttori sono di per sé dinamici

Operatori di classe

Operatore “is”

Sintassi:

object is class

Questo operatore esegue un controllo dinamico del tipo. In pratica permette di sapere se object è di tipo class o di una qualche classe figlia di class. E’ booleano (ritorna true o false). Se i tipi di object e class sono completamente slegati tra loro verrà segnalato un errore a tempo di compilazione

P.e.:

if ActiveControl is TEdit then TEdit(ActiveControl).SelectAll;

Operatore “as”

Sintassi:

object as class

Questo operatore esegue una conversione di tipo controllata. “Controllata” significa che object deve comunque essere o un’istanza di class o di una classe figlia di class o nil, altrimenti verrà sollevata un’eccezione. Se i tipi di object e class sono completamente slegati tra loro, verrà segnalato un errore a tempo di compilazione.

Un esempio di utilizzo:

with Sender as TButton do
begin

  Caption := '&Ok';

  OnClick := OkClick;

end;

La gestione delle eccezioni è un nuovo modo di gestire gli errori legato alla programmazione orientata agli oggetti.

· La gestione delle eccezioni viene attivata includendo la unit SysUtils

· Quando si verifica un errore viene interrotta la normale esecuzione del programma e si passa il controllo a un gestore dell’eccezione. Se il programmatore non ne fornisce uno viene attivato il gestore delle eccezioni standard

· Le eccezioni sono oggetti creati automaticamente al verificarsi di un errore e distrutti automaticamente senza l’intervento del programmatore. I campi e i metodi dell’eccezione forniscono informazioni sull’errore e permettono di gestirlo al meglio

· Tutte le eccezioni dovrebbero essere derivate dalla classe Exception o da una sua discendente. P.e.:

type

  EMathError = class(Exception);

  EInvalidOp = class(EMathError);

  EZeroDivide = class(EMathError);

Classi derivate possono aggiungere campi, metodi e proprietà in aggiunta alla gestione standard. P.e.:

type

 EInOutError = class(Exception)

    ErrorCode: Integer;

  end;

Sollevare un’eccezione

Un’eccezione può essere sollevata in seguito al verificarsi di un interrupt hardware (p.e. la divisione per 0) o essere invocata direttamente dal programma. Per sollevare un’eccezione si usa il comando raise. 

P.e.:

raise EMathError.Create;

· Il comando raise va invocato col costruttore dell’oggetto che descrive l’eccezione. Usare raise senza costruttore ha un significato differente

· L’oggetto costruito all’interno del comando raise non deve mai essere manualmente distrutto

· Invocando il comando raise si interrompe il normale flusso del programma. Il controllo viene trasferito al più interno gestore dell’eccezione che gestisce l’eccezione occorsa

Esempio:

function StrToIntRange(const S: string; Min, Max: Longint): Longint;

begin

  Result := StrToInt(S);  // StrToInt is declared in SysUtils

  if (Result < Min) or (Result > Max) then
    raise ERangeError.CreateFmt(

      '%d is not within the valid range of %d..%d', [Result, Min, Max]);

end;

Gestire un’eccezione

Le eccezioni vanno gestite con il costrutto di linguaggio try..except..end.

Sintassi:

try statements except exceptionBlock end
Le istruzioni in cui si può verificare l’eccezione da gestire vanno racchiuse tra try e except. Il codice di gestione va racchiuso tra except e end.

Il codice di gestione può essere:

· Una sequenza di istruzioni che gestisce ogni tipo di eccezione

· Una lista di gestori di eccezione eventualmente seguita da una clausola else
Un gestore di eccezione ha la forma:

on [identifier:] type do statement

dove identifier è opzionale ed esprime il nome dell’oggetto creato, type è il tipo di eccezione gestita e statement è l’azione specifica da intraprendere. La clausola else gestisce tutte le altre eccezioni non esplicitamente dichiarate. 

Un blocco di gestione eccezioni gestisce tutte le eccezioni se include la clausola else o se è una semplice sequenza di istruzioni, altrimenti gestisce solo i tipi di eccezione indicati. Se un’eccezione non è gestita dal blocco più interno si risale via via ai blocchi più esterni. Se nessun blocco gestisce l’eccezione viene utilizzato il gestore standard. In una lista di gestori, i primi hanno maggiore priorità degli ultimi 

Risollevare un’eccezione

E’ a volte necessario intercettare un’eccezione per liberare risorse allocate localmente senza però gestirla completamente ma permettendo anche a blocchi esterni di gestirla. In questi casi è necessario catturare l’eccezione e risollevarla. Questo viene fatto col comando raise. 

P.e.:

function GetFileList(const Path: string): TStringList;

var

  SearchRec: TSearchRec;

begin

  Result := TStringList.Create;

  try

    if FindFirst(Path, 0, SearchRec) = 0 then
    begin

      repeat

        Result.Add(SearchRec.Name)

      until FindNext(SearchRec) = 0;

      FindClose(SearchRec);

    end

  except

    Result.Free;

    raise;

  end;

end;

Costrutto try..finally

Sintassi:

try statementList1 finally statementList2 end
dove statementList1 e stamentList2 sono sequenze di istruzioni.

Il costrutto try..finally..end permette di assicurarsi che le istruzioni contenute in statementList2 vengano comunque eseguite, sia che si esca dal blocco di istruzioni statementList1 in maniera regolare sia che si sia verificata un’eccezione.

P.e.:

Reset(F);

try

 ...  // lavora con il file F

finally

  CloseFile(F);

end;

L’esecuzione di statementList2 è garantita anche in caso dell’uso di funzioni come Exit, Break, Continue all’interno di statementList1.

Classe Exception

La classe eccezione contiene 2 campi:

· Message contiene il messaggio da visualizzare

· HelpContext contiene il context ID per la documentazione on-line

Inoltre contiene diversi costruttori che permettono di inizializzare i 2 campi e di caricare il messaggio da un file risorse.

Routine di gestione eccezioni

In SysUtils sono definite molte routine per la gestione delle eccezioni. Ecco le principali:

	Funzione
	Descrizione

	ExceptAddr
	Ritorna l’indirizzo in cui si è verificata l’eccezione

	ExceptObject
	Ritorna l’oggetto associato all’eccezione

	ShowException
	Visualizza un messaggio di eccezione standard insieme all’indirizzo dove si è verificata l’eccezione

	SysErrorMessage
	Converte i codici di errore delle API in stringhe

	Win32Check
	Controlla il valore di ritorno di una funzione API e in caso di errore solleva l’eccezione appropriata

Discendenti di Exception

La VCL definisce molte classi che discendono da Exception, molte delle quali definite in SysUtils. Ecco un elenco non esaustivo:

· EMathError (errori aritmetica floating point)

· EInvalidArgument

· EInvalidOp


· EOverflow


· EUnderflow


· EZeroDivide


· EIntError (errori aritmetica intera)

· EDivByZero


· ERangeError


· EIntOverflow


· EInOutError (errori I/O)

· ErrorCode 2:
File not found.

· ErrorCode 3
Invalid file name.

· ErrorCode 4
Too many open files.

· ErrorCode 5
Access denied.

· ErrorCode 100
EOF.

· ErrorCode 101
Disk full.

· ErrorCode 106
Invalid input.

· EInvalidCast (uso improprio di as)
· EOutOfMemory (memoria esaurita)

· EVariantError (errore nell’uso dei variant)

· EWin32Error (errore di Windows)

· EDatabaseError (errore database)

· EDBEngineError (errore BDE)

